4 Gen_Event Behaviour
4 Поведение Gen_Event

This chapter should be read in conjunction with gen_event(3), where all interface functions and callback functions are described in detail. 
Эту главу следует читать вместе с документацией по gen_event(3), там более подробно описаны все интерфейсные функции и функции обратного вызова.
4.1 Event Handling Principles
4.1 Принципы обработки сообщений
In OTP, an event manager is a named object to which events can be sent. An event could be, for example, an error, an alarm or some information that should be logged. 
В OTP, менеджер событий(event manager) это именованый объект, которому можно посылать сообщения. Событие(event) может быть, например, ошибкой, нотификацией, или некоторой информацией, которую следует сохранить в логе. 

In the event manager, zero, one or several event handlers are installed. When the event manager is notified about an event, the event will be processed by all the installed event handlers. For example, an event manager for handling errors can by default have a handler installed which writes error messages to the terminal. If the error messages during a certain period should be saved to a file as well, the user adds another event handler which does this. When logging to file is no longer necessary, this event handler is deleted. 
В менеджере событий устанавливает ноль, один или несколько обработчиков событий. Когда менеджер событий получает уведомление о событии, событие обрабатывается всеми установленными обработчиками событий. Например, менеджер событий для обработки ошибок может иметь обработчик по умолчанию, который будет писать сообщения об ошибках на консоль. Если сообщения об ошибках должны быть сохранены в файл в течение определенного периода времени, пользователь добавляет еще один обработчик событий, котороый будет делать это.
An event manager is implemented as a process and each event handler is implemented as a callback module. 
Менеджер событий должен быть реализован как процесс, а каждый обработчик событий должен быть реализован как модуль обратного вызова.
The event manager essentially maintains a list of {Module, State} pairs, where each Module is an event handler, and State the internal state of that event handler. 
Менеджер событий содержит список пар {Module, State},где каждый Module это обработчик событий, а State – внутренне состояние обработчика событий.
4.2 Example
4.2 Пример
The callback module for the event handler writing error messages to the terminal could look like: 
Модуль обратного вызова для обработчика событий, реализующего вывод сообщений об ошибках на терминал, может выглядеть примерно так :
-module(terminal_logger).

-behaviour(gen_event).

-export([init/1, handle_event/2, terminate/2]).

init(_Args) ->

    {ok, []}.

handle_event(ErrorMsg, State) ->

    io:format("***Ошибка*** ~p~n", [ErrorMsg]),

    {ok, State}.

terminate(_Args, _State) ->

    ok.

The callback module for the event handler writing error messages to a file could look like: 
Модуль обратного вызова для обработчика событий, записывающего сообщения об ошибках в файл, может выглядеть так :
-module(file_logger).

-behaviour(gen_event).

-export([init/1, handle_event/2, terminate/2]).

init(File) ->

    {ok, Fd} = file:open(File, read),

    {ok, Fd}.

handle_event(ErrorMsg, Fd) ->

    io:format(Fd, "***Ошибка*** ~p~n", [ErrorMsg]),

    {ok, Fd}.

terminate(_Args, Fd) ->

    file:close(Fd).

The code is explained in the next sections. 
Этот код объясняется в следующих разделах.
4.3 Starting an Event Manager
4.3 Запуск менеджера событий
To start an event manager for handling errors, as described in the example above, call the following function: 
Для запуска менеджера событий для обработки ошибок, описанного в примере выше, вызывается следующая функция :
gen_event:start_link({local, error_man})

This function spawns and links to a new process, an event manager. 
Эта функция создает новый процесс, менеджер событий, а также создает связь с ним
The argument, {local, error_man} specifies the name. In this case, the event manager will be locally registered as error_man. 
Аргумент, {local, error_man} обозначает имя. В данном случае менеджер событий будет локально зарегистрирован как error_man.
If the name is omitted, the event manager is not registered. Instead its pid must be used. The name could also be given as {global, Name}, in which case the event manager is registered using global:register_name/2. 
Если имя будет пропущено, менеджер событий не будет зарегистрирован. В таком случае для общения с ним придется использовать его pid. Имя также может быть зарегистрировано как {global, Name}, в этом случае менеджер событий регистрируется с помощью функции global:register_name/2. 
gen_event:start_link must be used if the event manager is part of a supervision tree, i.e. is started by a supervisor. There is another function gen_event:start to start a stand-alone event manager, i.e. an event manager which is not part of a supervision tree. 
gen_event:start_link может быть использован, если менджер событий является частью дерева контроля, т.е. запускается контролером. Существует другая функция gen_event:start, для запуска менеджера событий в самостоятельном режиме, т.е. для менеджера событий, который не является частью дерева контроля.

4.4 Adding an Event Handler
4.4 Добавления обработчика событий
Here is an example using the shell on how to start an event manager and add an event handler to it: 
Ниже приеден пример использования оболочки для запуска менеджера событий и добавления к нему обработчика событий :
1> gen_event:start({local, error_man}).

{ok,<0.31.0>}

2> gen_event:add_handler(error_man, terminal_logger, []).

ok

This function sends a message to the event manager registered as error_man, telling it to add the event handler terminal_logger. The event manager will call the callback function terminal_logger:init([]), where the argument [] is the third argument to add_handler. init is expected to return {ok, State}, where State is the internal state fo the event handler. 
Эта функция посылает сообщение менеджеру событий, зарегистрированному как error_man, приказывая ему добавить обработчик событий terminal_logger.Менеджер событий вызовет функцию обратного вызова terminal_logger:init([]),в которой аргумент [] является третьим аргументом для функции add_handler. init должна вернуть {ok, State}, где State – внутреннее состояние обработчика событий. 
init(_Args) ->

    {ok, []}.

Here, init does not need any indata and ignores its argument. Also, for terminal_logger the internal state is not used. For file_logger, the internal state is used to save the open file descriptor. 
В данном случае, функции init не нужны какие-либо внутренние данные, и она игнорирует аргумент. Также внутреннее состояние не используетя для функции terminal_logger. Для функции file_logger, внутреннее состояние используется для сохранения дескриптора открытого файла. 

init(_Args) ->

    {ok, Fd} = file:open(File, read),

    {ok, Fd}.

4.5 Notifying About Events
4.5 Уведомление о событиях
3> gen_event:notify(error_man, no_reply).

***Ошибка*** no_reply

ok

error_man is the name of the event manager and no_reply is the event. 
error_man это имя менеджера событий, а no_reply это событие
The event is made into a message and sent to the event manager. When the event is received, the event manager calls handle_event(Event, State) for each installed event handler, in the same order as they were added. The function is expected to return a tuple {ok, State1}, where State1 is a new value for the state of the event handler.
Событие преобразовывается в сообщение и посылается менеджеру событий. Когда событие получено, менеджер событий вызывает handle_event(Event, State) для каждого установленного обработчика событий, в том порядке, в каком они добавлялись. Эта функция должна вернуть кортеж {ok, State1}, где State1 – новое состояние обработчика событий.
In terminal_logger: 
В terminal_logger-е: 

handle_event(ErrorMsg, State) ->

    io:format("***Error*** ~p~n", [ErrorMsg]),

    {ok, State}.

В file_logger-е: 

handle_event(ErrorMsg, Fd) ->

    io:format(Fd, "***Error*** ~p~n", [ErrorMsg]),

    {ok, Fd}.

4.6 Deleting an Event Handler
4.6 Удаление обработчика событий
4> gen_event:delete_handler(error_man, terminal_logger, []).

ok

This function sends a message to the event manager registered as error_man, telling it to delete the event handler terminal_logger. The event manager will call the callback function terminal_logger:terminate([], State), where the argument [] is the third argument to delete_handler. terminate should be the opposite of init and do any necessary cleaning up. Its return value is ignored. 
Функция посылает сообщение менеджеру событий, зарегистрированному как error_man, приказывая ему удалить обработчик событий terminal_logger. Менджер событий вызовает функцию обратного вызова terminal_logger:terminate([], State), где аргумент [] это третий аргумент функции delete_handler. terminate должен быть «антонимом» init и выполнять необходимую очистку ресурсов. Значение, возвращаемое этим методом, игнорируется. 
For terminal_logger, no cleaning up is necessary: 
Для terminal_logger-а, освобождение ресурсов необязательно: 

terminate(_Args, _State) ->

    ok.

For file_logger, the file descriptor opened in init needs to be closed: 
Для file_logger,файл, открытый в init должен быть закрыт: 

terminate(_Args, Fd) ->

    file:close(Fd).

4.7 Stopping
4.7 Завершение работы
When an event manager is stopped, it will give each of the installed event handlers the chance to clean up by calling terminate/2, the same way as when deleting a handler. 

4.7.1 In a Supervision Tree
4.7.1 В дереве контроля
If the event manager is part of a supervision tree, no stop function is needed. The event manager will automatically be terminated by its supervisor. Exactly how this is done is defined by a shutdown strategy set in the supervisor. 
Если менджер событий является частью дерева контроля, функция завершения работы не требуется. Менеджер событий будет автоматически завершен контролером. Как именно это будет сделано, определяется стратегией завершения (shutdown strategy), установленной в контролере.
4.7.2 Stand-Alone Event Managers
4.7.2 Самостоятельные менеджеры событий
An event manager can also be stopped by calling: 
Менеджер событий может быть остановлен следующим вызовом :
> gen_event:stop(error_man).

ok







